6701 ODE Notes/ Final Exam 2020

Plotting a vector field

In[1]:=
```
VectorPlot[{(1 / 2) x (1 - x / 5 - y), (1 / 10) y (-1 + 2 x - 3 y / 10)}, {x, 0, 2}, {y, 0, 2}]
```

Out[1]=



In[2]:=
```
interest = VectorPlot[{(1 / 2) x (1 - x / 5 - y), (1 / 10) y (-1 + 2 x - 3 y / 10)} /
    Norm[{(1 / 2) x (1 - x / 5 - y), (1 / 10) y (-1 + 2 x - 3 y / 10)}], {x, 0, 2}, {y, 0, 2}]
```

Power: Infinite expression $\frac{1}{0.}$ encountered.

Infinity: Indeterminate expression 0. ComplexInfinity encountered.

Infinity: Indeterminate expression 0. ComplexInfinity encountered.

Out[2]=



Equilibrium Point

In[3]:=
```
field[x_, y_] = {(1 / 2) x (1 - x / 5 - y), (1 / 10) y (-1 + 2 x - 3 y / 10)}
```

Out[3]= $\left\{\frac{1}{2} x \left(1 - \frac{x}{5} - y\right), \frac{1}{10} \left(-1 + 2 x - \frac{3 y}{10}\right) y\right\}$

In[4]:= $\text{Solve}\left[\left\{1 - \frac{x}{5} - y == 0, -1 + 2 x - \frac{3 y}{10} == 0\right\}, \{x, y\}\right]$

Out[4]= $\left\{\left\{x \to \frac{65}{103}, y \to \frac{90}{103}\right\}\right\}$

Linearization

In[5]:=
```
field[x, y][[1]]
```

Out[5]= $\frac{1}{2} x \left(1 - \frac{x}{5} - y\right)$

In[6]:=
```
matrix[x_, y_] =
  {{D[field[x, y][[1]], x], D[field[x, y][[1]], y]}, {D[field[x, y][[2]], x], D[field[x, y][[2]], y]}}
```

Out[6]= $\left\{\left\{-\frac{x}{10} + \frac{1}{2}\left(1 - \frac{x}{5} - y\right), -\frac{x}{2}\right\}, \left\{\frac{y}{5}, \frac{1}{10}\left(-1 + 2 x - \frac{3 y}{10}\right) - \frac{3 y}{100}\right\}\right\}$

In[7]:=
```
matrix[65 / 103, 90 / 103]
```

Out[7]= $\left\{\left\{-\frac{13}{206}, -\frac{65}{206}\right\}, \left\{\frac{18}{103}, -\frac{27}{1030}\right\}\right\}$

In[8]:=
```
Det[matrix[65 / 103, 90 / 103] - lambda {{1, 0}, {0, 1}}]
```

Out[8]= $\frac{117}{2060} + \frac{46\,\text{lambda}}{515} + \text{lambda}^2$

In[9]:=
```
Solve[Det[matrix[65 / 103, 90 / 103] - lambda {{1, 0}, {0, 1}}] == 0, lambda]
```

Out[9]= $\left\{\left\{\text{lambda} \to \frac{-46 - i\sqrt{58139}}{1030}\right\}, \left\{\text{lambda} \to \frac{-46 + i\sqrt{58139}}{1030}\right\}\right\}$

Numerical Solution

The following lines contain what may be the most practically useful information presented in this course.

In[10]:=
```
soln[xzero_, yzero_] :=
  NDSolve[{odex '[t] == field[odex[t], odey[t]][[1]], odey '[t] == field[odex[t], odey[t]][[2]],
    odex[0] == xzero, odey[0] == yzero}, {odex, odey}, {t, 0, 10}]
```

In[11]:=
```
soln[1, 1]
```

Out[11]= {{odex → InterpolatingFunction[

Domain: {{0., 10.}}
Output: scalar
],

odey → InterpolatingFunction[

Domain: {{0., 10.}}
Output: scalar
]}}

In[12]:=
```
onesoln[t_] = {odex[t] /. soln[1, 1][[1]], odey[t] /. soln[1, 1][[1]]}
```

Out[12]=
```
{InterpolatingFunction[
```

```
Domain: {{0., 10.}}
Output: scalar
][t],
 InterpolatingFunction[
```

```
Domain: {{0., 10.}}
Output: scalar
][t]}
```

In[13]:=
```
Show[interest, ParametricPlot[onesoln[t], {t, 0, 10}, PlotStyle → {Thick, Red}]]
```

Out[13]=



In[14]:=
```
solnlong[xzero_, yzero_] :=
 NDSolve[{odex'[t] == field[odex[t], odey[t]][[1]], odey'[t] == field[odex[t], odey[t]][[2]],
   odex[0] == xzero, odey[0] == yzero}, {odex, odey}, {t, 0, 100}]
```

In[15]:=
```
onesolnlong[t_] = {odex[t] /. solnlong[1, 1][[1]], odey[t] /. solnlong[1, 1][[1]]}
```

Out[15]=
```
{InterpolatingFunction[
```

```
Domain: {{0., 100.}}
Output: scalar
][t],
 InterpolatingFunction[
```

```
Domain: {{0., 100.}}
Output: scalar
][t]}
```

In[16]:=
```
Show[interest, ParametricPlot[onesolnlong[t], {t, 0, 50}, PlotStyle → {Thick, Red}]]
```

Out[16]=



In[17]:=
```
Animate[Show[interest,
  ParametricPlot[onesolnlong[t], {t, 0, end}, PlotStyle → {Thick, Red}]], {end, 0.1, 100}]
```

Out[17]=



Another Equilibrium Point

In[18]:=
```
field[5, 0]
```
Out[18]= $\{0, 0\}$

In[19]:=
```
matrix[5, 0]
```
Out[19]= $\left\{\left\{-\frac{1}{2}, -\frac{5}{2}\right\}, \left\{0, \frac{9}{10}\right\}\right\}$

In[20]:=
```
matrix[0, -10/3]
```
Out[20]= $\left\{\left\{\frac{13}{6}, 0\right\}, \left\{-\frac{2}{3}, \frac{1}{10}\right\}\right\}$